Valve32

In[115]:=
```
<< C:\\Hopsan\Compgen\CompgenNG.mx
```

In[116]:=
```
path = ToFileName[{"H:", "PettersDropbox", "Dropbox",
    "HopsanComponents", "PneumaticDevelop", "PneumaticComponents"}]
```

Out[116]= H:\PettersDropbox\Dropbox\HopsanComponents\PneumaticDevelop\PneumaticComponents\

In[117]:=
```
path = ToFileName[{"C:", "Users", "petkr14", "Dropbox",
    "HopsanComponents", "PneumaticDevelop", "PneumaticComponents"}]
```

Out[117]= C:\Users\petkr14\Dropbox\HopsanComponents\PneumaticDevelop\PneumaticComponents\

In[118]:=
```
domain = "Pneumatic";
displayName = "Valve32";
brief = "Pneumatic 32-valve";
componentType = "ComponentQ";
author = "Petter Krus <petter.krus@liu.se>";
affiliation = "Division of Fluid and Mechatronic Systems, Linköping University";
SetFilenames[path, domain, displayName];
ResetComponentVariables[];
Date[]
```

Out[126]= {2020, 7, 29, 13, 54, 3.1178489}

In[127]:=
```
eps =.; R =.;
```



In[128]:=
```
inputParameters = {
    {Cd, 0.65, double, "", "Discharge coefficient"},
    {R, 287., double, "J/Kg K", "Gas constant"},
    {cv, 718, double, "J/Kg K", "heatcoeff"},
    {eps, 0.02, double, "", "Linearisation coeff"},
    {x0, 0.1, double, "", "Relative overlap"},
    {A1max, 1. * 10^-5, double, "m2", "Max opening area"},
    {A3max, 1. * 10^-5, double, "m2", "Max opening area"}
   };
```

In[129]:=
```
inputVariables = {
    {xin, 1, double, "", "Input signal 0<xin<1"}
   };
```

```
In[130]:= outputVariables = {
            {qm12Pos, 0., double, "kg/s", "Internal variable"},
            {qm12Neg, 0., double, "kg/s", "Internal variable"},
            {qm32Pos, 0., double, "kg/s", "Internal variable"},
            {qm32Neg, 0., double, "kg/s", "Internal variable"},
            {Ng32e, 0., double, "", "Internal variable"},
            {Ng12e, 0., double, "", "Internal variable"}
           };

In[131]:= nodeConnections = {
            PneumaticQnode[2, 100000., "fluid port 2"],
            PneumaticQnode[1, 100000., "fluid port 1"],
            PneumaticQnode[3, 100000., "fluid port 3"]
           };

In[132]:= 0.01 × 2 Pi .001

Out[132]= 0.0000628319
```

The system of equations

```
In[133]:= xine = 2 xin - 1;
```

The valve areas are limited between 0 and A1max and A3max respectively .

```
In[134]:= A12 = A1max limit[(xine - x0)/(1 - x0), 0, 1];

          A32 = A3max limit[(-xine - x0)/(1 - x0), 0, 1];
```

The flow at inlet and outlet are equal but with opposite sign.

```
In[136]:= Ng1 = 1;

In[137]:= Ng12pos = (signedSquareL[((p2/p1)^(2/kappa) - (p2/p1)^((kappa+1)/kappa))/Ndenom, eps]);

In[138]:= Ng12neg = (signedSquareL[((p1/p2)^(2/kappa) - (p1/p2)^((kappa+1)/kappa))/Ndenom, eps]);

In[139]:= Ng12 := onPositive[p1 - p2] (onPositive[p2/p1 - crit] Ng12pos + onNegative[p2/p1 - crit] Ng1) +
              onNegative[p1 - p2] (onPositive[p1/p2 - crit] Ng12neg + onNegative[p1/p2 - crit] Ng1);

In[140]:= Ng32pos = (signedSquareL[((p2/p3)^(2/kappa) - (p2/p3)^((kappa+1)/kappa))/Ndenom, eps]);

In[141]:= Ng32neg = (signedSquareL[((p3/p2)^(2/kappa) - (p3/p2)^((kappa+1)/kappa))/Ndenom, eps]);
```

In[142]:=
```
Ng32 := onPositive[p3 - p2] (onPositive[p2/p3 - crit] Ng32pos + onNegative[p2/p3 - crit] Ng1) +
    onNegative[p3 - p2] (onPositive[p3/p2 - crit] Ng32neg + onNegative[p3/p2 - crit] Ng1);
```

Equations

In[143]:=
```
localExpressions = {
    kappa == 1 + R/cv,
    Kg == Sqrt[2^((kappa+1)/(kappa-1)) kappa (1/(kappa+1))^((kappa+1)/(kappa-1)) / R],
    Ndenom == 2^((kappa+1)/(kappa-1) - 1) (kappa - 1) (1/(kappa + 1))^((kappa+1)/(kappa-1)),
    crit == 2^(kappa/(kappa-1)) (1/(kappa + 1))^(kappa/(kappa-1)),
    cp == cv + R
   };
```

In[144]:=
```
dE12 = qm12 cp (onNegative[qm12] T2 + onPositive[qm12] T1);
dE32 = qm32 cp (onNegative[qm32] T2 + onPositive[qm32] T3);

qm12 = (onPositive[p1 - p2] qm12Pos - onNegative[p1 - p2] qm12Neg);
qm32 = (onPositive[p3 - p2] qm32Pos - onNegative[p3 - p2] qm32Neg);
```

In[148]:=
```
systemEquationsDA = Simplify[{
     qm12Pos == p1 Cd A12 Kg Ng12/Sqrt[T1],
     qm12Neg == p2 Cd A12 Kg Ng12/Sqrt[T2],
     qm32Pos == p3 Cd A32 Kg Ng32/Sqrt[T3],
     qm32Neg == p2 Cd A32 Kg Ng32/Sqrt[T2],
     dE2 == dE12 + dE32,
     dE1 == -dE12,
     dE3 == -dE32
        }];
```

Boundarys

In[149]:=
```
systemBoundaryEquations = {
    p2 == (c2 + Zc2 dE2),
    p1 == (c1 + Zc1 dE1),
    p3 == (c3 + Zc3 dE3)
   };
```

Independent Variables

In[150]:=
```
systemVariables = {qm12Pos, qm12Neg, qm32Pos, qm32Neg, dE2, dE1, dE3, p2, p1, p3};
```

Expressions

The inlet flow is calculated as the outlet flow with reversed sign.

In[151]:=
```
expressions = {
    qm2 == qm12 + qm32,
    qm1 == -qm12,
    qm3 == -qm32,
    Ng32e == Ng32,
    Ng12e == Ng12
   };
```

In[152]:=
```
Compgen[file]
```